APPLICATION NOTE

EIE/AN91006
A/D conversion with P83CL410
PCF1252-x

Jun 1991

Philips
Semiconductors

PHILIPS

Author: Th. v. Daele, Product Concept & Application Laboratory, Eindhoven, The Netherlands

1. INTRODUCTION

With an 83CL410 microcontroller and a PCF1252-x reset circuit, it is possible to make a software driven A-to-D converter. In this application note, an example is described where an 83CL410 measures its own supply voltage. The resolution of the measurement is 0.1V. The program example also refers to this application.

Chapter 2 describes the algorithm of the conversion. In chapter 3 the example with 83CL410/PCF1252 is described. Both hardware and software of this example are explained.

References:

– 80C51-based 8-bit microcontrollers; Data Handbook IC20

– PCF1252-x data sheet

2. A-TO-D CONVERSION PRINCIPLE

The basic principle of the conversion is to convert the voltage to a time measurement. A microcontroller without an on-chip A-to-D converter cannot measure voltages directly, but if converted to a time measurement, this can be done by software or with the help of an on-chip timer/count.

Figure 1 shows the basic circuit to do the conversion. The circuit consists of an integrator circuit, a voltage reference and an analog input switch S. The analog input switch is controlled by the microcontroller. The integrator is built around a comparator whose output is connected to a microcontroller input.



Figure 1.

Before the measurement is started, the analog switch connects the integrator input to ground, so that the integration capacitor is fully discharged. The measurement is started by connecting the integrator input to the unknown voltage. The integration capacitor will charge up. When the non-inverting input exceeds the reference voltage, the comparator output will switch from LOW to HIGH.

The time between starting the measurement and the moment that the comparator output becomes HIGH, is measured by the microcontroller and is:

$$T_1 = -RC.ln\frac{V_x - V_{ref}}{V_x}$$

The charging continues until the capacitor is fully charged. Then the integrator input is grounded via the analog switch. The integration capacitor discharges while the comparator output is HIGH. When the input voltage becomes lower than the reference voltage, the comparator output becomes LOW again. The time between the start of the discharging and the moment that the comparator output becomes LOW, is measured by the microcontroller and is:

$$T_2 = -RC.ln\frac{V_{ref}}{V_x}$$

When the microcontroller uses the ratio between T_1 and T_2, the result becomes independent of the values R and C. The resulting ratio can then be used as a pointer to a look-up table that may contain an indication for the measured parameter or display data.

3. EXAMPLE OF CONVERSION PRINCIPLE

3.1 Hardware

In the example, an application is used where a microcontroller measures its own supply voltage. The supply voltage is generated by solar cells, so power consumption should be minimized. The measured voltage is shown on an LCD display.

An 83CL410 is used as microcontroller. This controller is an 80C51 family member. Compared with a standard 80C51, it has the following extra features:

- Wide supply voltage range of 1.8V to 6V.
- Wide operating frequency range of 32kHz to 20MHz with internal oscillator. With external oscillator there are no limitations on the lower frequency limit.
- Byte I^2C interface instead of UART.
- 8 extra external interrupt inputs on P1. The interrupt level is programmable. These interrupts can terminate the power-down mode.
- 3 mask programmable I/O port configurations.
  These configurations are:
  - Standard quasi-bidirectional I/O
  - Open-drain output, standard input
  - Push-pull output, no input
- I/O port levels after RESET are mask programmable.

Of these features, only the byte I^2C interface is not used.

The PCF1252 is used for monitoring the power supply voltage and generating a reset pulse when the supply drops too much. Several versions of PCF1252 are available, every one with its own trip voltage. The PCF1252 also has an unused comparator. This comparator is used for the integrator circuit. The inverting input is internally connected to a 1.3V reference.

Figure 2 shows the A-to-D conversion part of the circuit.



Figure 2.

R1 and C1 determine the time-constant of the circuit. The input of the integrator is connected to P1.1 of the microcontroller. This port line has a push-pull configuration. It is used as the analog switch shown in Figure 1. When this output line is made HIGH, the integrator is connected to the supply voltage. This is the voltage to be measured. The voltages on the non-inverting input and the output are shown in Figure 2. The output of the comparator is LOW, and the capacitor will be charged. When the voltage at the non-inverting input becomes higher than 1.3V, the comparator output will become HIGH. This HIGH level will cause an interrupt on P1.0 (IL2=1). The voltage on the inverting input will want to rise to V_{CC} + 1.3V. this voltage is limited by 2 external diodes. R2 is a current limit resistor.

Next step is that P1.1 becomes LOW, so that the capacitor will be discharged. The P1.0 input is now programmed to generate an interrupt on a LOW level (IL2=0). This will happen when the voltage on the inverting input has dropped below 1.3V.

At this point, both charge and discharge times are known, and the supply voltage can be calculated. In this circuit, the PCF1252-9 is used, which will generate a reset pulse when the supply voltage is lower than 2.55V. This limits the lower end of the supply voltage range. However, the 83CL410 is able to go as low as 1.8V.

To minimize power consumption, the lowest frequency is used where the internal oscillator can still be sued. this is 32.768kHz. This parameter, together with the resolution that must be met, determine the minimum value of R1.C1. In this example, a resolution of 0.1V is achieved. The software is made to handle voltage measurements from 1.8V to 5V.

To determine the minimum value of R1C1, the measurement of 5.0V and 4.95V is important because here the distance between measurements of T_1 and T_2 is minimal.

$V_x = 5.0V$: $T_1 = 0.301RC$
$T_2 = 1.347RC$
$T_2/T_1 = 4.437$

$V_x = 4.95V$: $T_1 = 0.304RC$
$T_2 = 1.337RC$
$T_2/T_1 = 4.398$

$V_x = 4.9V$: $T_1 = 0.308RC$
$T_2 = 1.327RC$
$T_2/T_1 = 4.308$

Because of latency in the interrupt routine, a deviation Δt may occur in T_1 and T_2. The following conditions must be met:

$V_x = 5.0V$: $$\frac{T_2}{T_1} \quad \frac{1.347RC \pm \Delta t}{0.301RC \pm \Delta t} \geq 4.398$$

$V_x = 4.9V$: $$\frac{T_2}{T_1} \quad \frac{1.327RC \pm \Delta t}{0.308RC \pm \Delta t} \geq 4.398$$

In this example, there is an uncertainty of 2 machine cycles when measuring T_1 and T_2. Given an XTAL frequency of 32.768kHz, the minimum value of RC is 0.172. The complete circuit diagram is shown of the example with the LCD connections. Since the LCD connections are only outputs, the push-pull configuration is used for this.


Vss=0
Vdd=p40
U1
P83CL410HFT
EA
X1
X2
RESET
PSEN
ALE
P0.6
P0.7
P1.0
P1.1
P1.2
P1.3
P1.4
P1.5
P1.6
P1.7
P3.0
P3.1
P3.2
P3.3
P3.4
P3.5
P3.6
P3.7
P2.0
P2.1
P2.2
P2.3
P2.4
P2.5
P2.6
P2.7
P0.0
P0.1
P0.2
P0.3
P0.4
P0.5
U3
LTD227R-12
LOBAT
COM
VCC
32.768kHz
X1
R1
4.7M
C1
10pF
C2
2uF
D1
BAS45L
D2
BAS45L
R2
1k
R3
100k
C3
47nF
U2
PCF1252-9T
COMIN
CT
SELECT
COMOUT
POWF
RESET
D3
BPY63P
D12
BPY63P
C4
22uF
C5
100nF
C6
100nF

3.2 Software

In chapter 4 the listing of the program is shown.

After RESET the initialization takes place from L.45 to L.70.

The following timers 83CL410 are used:

Timer_0:	Used to measure the T1 and T2 (mode_1).
Timer_1:	Used to generate a timer interrupt for LCD polarity reverse (mode_2). This is the auto-reload mode. Every 10ms an interrupt is generated.

3 interrupts are used:

INT2:	External interrupt for the A-to-D conversion. Priority level–1.
INT4:	Interrupt from PCD1252-9 when supply voltage is smaller than 2.55V. Priority level_1.
Timer_1:	Interrupt for LCD polarity reverse. Priority level_0.

The measurement is started at L74. The interrupt level of INT2 is set to ‘1’ and time_0 is started to measure T1. P1.1 is set, so that the supply-voltage is connected to the integrator input. While the integration capacitor is charging, the 83CL410 is in IDLE-mode to reduce power consumption. This part of the measurement is identified by ADC_Status=1. When INT2 is generated, the interrupt routine is entered (from L.181). Timer_0 is read and its value is stored in R4 (MSB) and R5 (LSB). The interrupt level of INT2 is cleared to ‘0’ for the second part of the measurement. After clearing the interrupt flag, the main program is entered again at L.85. a delay is entered at this point so that the integration capacitor can be discharged sufficiently from V_x + 1.3V to V_x. The delay is 32 time-outs of timer_1. During this delay, the controller is in IDLE mode. After the delay, P1.0 is cleared to discharge the integration capacitor. Timer_0 is started to measure T2. Again the microcontroller enters the IDLE mode until INT2 is generated. In the interrupt routine, the interrupt level is now set to ‘1’ again for the next A-to-D conversion. In the main program, T2 is copied to R6 (MSB) and R7 (LSB) (L.96 and L97).

Now that T1 and T2 are known, the measured voltage can be determined. This is done by calculating the ratio between T1 and T2, and converting this to a pointer which points to a table with the segment data of the LCD display. In the calculation, the ratio between T1 and T2 should be equal or greater than 1. If T2<T1 (L.107) then the flag ‘gr_2V6’ will be cleared (L1.65) and [R4.R5] and [R6.R7] are exchanged. This point is met when the input voltage is 2.55V.

Normally, voltages < 2.55V will not be measured, because by then the PCF1252-9 has generated an interrupt on INT4. This means a power-failure and the only way to start the measurement again, is that the PCF1252-9 resets the microcontroller. The INT4 routine is from L.227 .. L.237). The program, however, contains the conversion table for input voltages as low as 1.8V.

The pointer to the table (stored in [R6.R7]) is calculated by: [R6.R7] = 16*[R6.R7] / [R4.R5]. The division is done by the subroutine ‘_sdivi’. This is a library function contained in the C-library of the BSO/Tasking C-compiler package (type number OM41326). This library must be linked to this program.

Of the result, only R7 is relevant. The segment data contains 2 bytes, so pointer R7 is multiplied by 2 (L.134). There are 2 segment tables: one for voltages > 2.55V (L.247 .. L.274) and one for voltages < 2.55V (L.284 .. L.304). DPTR is used as pointer for this table and is calculated by adding the table base (L.129 .. L.132) and R7 (L.134 .. L.137). The following table shows the relation between the measured voltage, 16*T1/T2 or 16*T2/T1, the pointer and the segment data.

Input voltage > 2.55V				
Input Voltage	16*T2/T1	Table addresses	Segment data	Display
4.95	70.4	140 .. 147	0x6d,0xbf	5.0V
4.85	67.5	136 .. 139	0x66,0xef	4.9V
4.75	64.83	130 .. 135	0x66,0xff	4.8V
4.65	62.18	124 .. 129	0x66,0x87	4.7V
4.55	59.57	120 .. 123	0x66,0xfd	4.6V
4.45	56.98	114 .. 119	0x66,0xed	4.5V
4.35	54.43	108 .. 113	0x66,0xeb	4.4V
4.25	51.91	104 .. 107	0x66,0xcf	4.3V
4.15	49.42	98 .. 103	0x66,0xdb	4.2V
4.05	46.96	94 .. 97	0x66,0x86	4.1V
3.95	44.54	90 .. 93	0x66,0xbf	4.0V
3.85	42.14	84 .. 89	0x4f,0xef	3.9V
3.75	39.82	80 .. 83	0x4f,0xff	3.8V
3.65	37.51	76 .. 79	0x4f,0x87	3.7V
3.55	35.24	70 .. 75	0x4f,0xfd	3.6V
3.45	33.02	66 .. 69	0x4f,0xed	3.5V
3.35	30.83	62 .. 65	0x4f,0xeb	3.4V
3.25	28.7	58 .. 61	0x4f,0xcf	3.3V
3.15	26.6	54 .. 57	0x4f,0xdb	3.2V
3.05	24.56	48 .. 53	0x4f,0x86	3.1V
2.95	22.56	46 .. 47	0x4f,0xbf	3.0V
2.85	20.62	42 .. 45	0xdb,0xef	2.9V LOBAT
2.75	18.73	38 .. 41	0xdb,0xff	2.8V LOBAT
2.65	16.89	34 .. 37	0xdb,0x87	2.7V LOBAT
2.55	15.12	32 .. 34	0xdb,0xfd	2.6V LOBAT

Input voltage < 2.55V				
Input Voltage	16*T1/T2	Table addresses	Segment data	Display
2.55	16.92	32 .. 37	0xdb,0xed	2.5V LOBAT
2.45	19.09	38 .. 43	0xdb,0xeb	2.4V LOBAT
2.35	21.77	44 .. 49	0xdb,0xcf	2.3V LOBAT
2.25	25.15	50 .. 57	0xdb,0xdb	2.2V LOBAT
2.15	29.51	58 .. 69	0xdb,0x86	2.1V LOBAT
2.05	25.32	70 .. 85	0xdb,0xbf	2.0V LOBAT
1.95	43.35	86 .. 109	0x86,0xef	1.9V LOBAT
1.85	54.96	110 .. 119	0x86,0xff	1.8V LOBAT

Now the data can be displayed on the LCD display. When writing data to the LCD, the timer_1 interrupt is disabled. The value of output LCD_COM (controlled by timer_1 interrupt routine) determines whether the data must be written inverted to the LCD or not. The data is written non-inverted from L.147 .. L.150, inverted from L.153 .. L.158. When the data is written, timer_1 interrupt is enabled again, and a new A-to-D conversion is started at label Start_ADC.

From L.210 .. L219 the timer_1 interrupt routine is shown. In this routine the segment lines to the LCD display are inverted (on P2 and P3). Also LCD_COM is inverted, which is the COMMON pin of the display.

On L218 the value Dis_tim is incremented. This is for the delay between the measurement of T1 and T2 (see L.85).

4. SOFTWARE LISTING

```
TSW–ASM51 V3.0b Serial #00052252  Main program
PAGE   1

LOC   OBJ          LINE  SOURCE

                      1  $TITLE(Main program)
                      2  $DEBUG
                      3  $
                      4  # 1 ”C:\Tools\Tasking\ASM51\Include\8051\Reg410.Inc”
                      5
;***************************************************************************
                      6  ;Timer_0 is used for A/D conversion in combination with PCF1252
                      7  ;Timer_1 is used as a time base for LCD switching
                      8  ;INT2 is used a input for A/D conversion
                      9  ;INT4 is used a power–failure input from PCF1252
                     10  ;All timing related to Xtal=32.768kHz
                     11
;***************************************************************************
                     12
0091                 13                 Anal_in   EQU   91h     ;P1.1 is voltage switch
0095                 14                 LCD_COM   EQU   95h     ;P1.5 is COM of LCD display
00B0                 15                 Digit_1   EQU   0B0h    ;P3 is first digit of LCD
                                                                ;display
00A0                 16                 Digit_2   EQU   0A0h    ;P2 is second digit of LCD
                                                                ;display
                     17
                     18
;***************************************************************************
                     19
                     20
                     21                 Flags   segment Bit
                     22                 Stack   segment Data
                     23                 Main    segment Code Inblock
                     24                 Table_1 segment Code Page
                     25                 Table_2 segment Code Page
                     26                 ADC     segment Code
                     27                 LCD     segment Code
                     28                 Power_F segment Code
                     29
––––                 30                 RSEG Flags
0000:           R    31  ADC_Status:    DBIT 1          ;Status flag AD converter
0001:                32  Gr_2v6:        DBIT 1          ;Flag indicating that V<2.55V
                     33
––––                 34                 REG Stack
0000:           R    35  Stk_St:        DS 20           ;Define stack
                     36
REG END              37                 Dis_tim set r2  ;Timer for cap. discharge from Vdd+1.3
                                                        ;to Vdd
                     38
                     39
```

```
TSW–ASM51 V3.0b Serial #00052252  Main program
PAGE   2

LOC   OBJ         LINE  SOURCE

;************************************************************************
                   40
----               41                 CSEG AT 00      ;Reset vector
0000: 020000   R   42                 ljmp Start
                   43
----               44                 RSEG Main
0000: C291     R   45  Start:         clr Anal_in     ;Ground analog input
0002: 7581FF   R   46                 mov sp,#Stk_st–1 ;Initialize stack pointer
0005: C200     R   47                 clr ADC_Status  ;Initialize ADC status
                   48
0007: 75F805       49                 mov ipl,#05h    ;INT2 is priority level 1  (Analog
                                                      ;measurem.)
                   50                                 ;INT4 is priority level 1  (Power fail)
                   51                                 ;Timer_1 interrupt level 0 (LCD switch)
000A: 53E9FB       52                 anl ix1,#0fbh   ;INT4 on low level
000D: D2EA         53                 setb ex4        ;Enable INT4
000F: D2AF         54                 setb ea         ;Global enable interrupts
                   55
0011: 758921       56                 mov tmod,#21h   ;Timer_1 mode (LCD switch)
                   57                                 ;     8 bit auto reload mode
                   58                                 ;Timer_0 mode (A/D conversion)
                   59                                 ;     16 bit timer
0014: 758DF5       60                 mov th1,#0f5h   ;Delay for discharging cap’s (+/– 1sec)
0017: 758B50       61                 mov tl1,#050h
001A: D28E         62                 setb tr1
001C: 308FFD       63                 jnb tf1,$
001F: C28F         64                 clr tf1
0021: C28E         65                 clr tr1
                   66
0023: 758BE1       67                 mov tl1,#0e1h   ;LCD switch rate: 10ms
0026: 758DE1       68                 mov th1,#0e1h
0029: D28E         69                 setb tr1        ;Start timer_1
002B: D2AB         70                 setb et1        ;Enable timer_1 interrupt
                   71
                   72  ;Measurement of supply voltage
002D: 43E901       73                 orl ix1,#01H    ;INT2 on ‘1’ input level
0030:              74  Start_ADC:
0030: 758CFF       75                 mov th0,#0FFh   ;Load timer_0
0033: 758AFB       76                 mov tl0,#0fBh
0036: D28C         77                 setb tr0        ;Start timer_0
0038: D291         78                 setb Anal_in    ;Start raising part of measurement
003A: D2E8         79                 setb ex2        ;Enable INT2
003C: 438701       80  ID_1:          orl pcon,#01    ;Go to idle mode
003F: 3000FA   R   81                 jnb ADC_Status,Id_1;Wait till first part of measurement
                                                      ;is handled
                   82                                 ;Exit from idle mode could be from LCD
                                                      ;interrupt
```

```
TSW–ASM51 V3.0b Serial #00052252  Main program
PAGE   3

LOC   OBJ         LINE  SOURCE

0042: 7A00                              mov Dis_tim,#00 ;Discharge delay: Vdd+1.3...Vdd
0044: 438701            Id_2:           orl pcon,#01    ;Go to idle mode
0047: BA20FA                            cjne Dis_tim,#20h,Id_2 ;Delay: 32 LCD_time_outs
                                                        ;If not equal: idle
004A: 758CFF                            mov th0,#0FFh   ;Load timer_0
004D: 758AFB                            mov tl0,#0FBh
0050: D28C                              setb tr0        ;Start timer_0
0052: C291                              clr Anal_in     ;Start second part of measurement is
                                                        ;handled
0054: 438701            Id_3:           orl pcon,#01    ;Go to idle mode
0057: 2000FA   R                        jb ADC_Status,Id_3 ;Wait till measurement is finished
005A: C2E8                              clr ex2         ;Disable INT2
005C: AE8C                              mov r6,th0      ;Get timer data of second part of
                                                        ;measurement
005E: AF8A                              mov r7,tl0      ;T2
                                                        ;Process data
                                                        ;If T1>T2 then [R6.R7] and [R4.R5] must
                                                        ;be exchanged. Also other segment table
                                                        ;must be used.
0060: D201     R                        setb gr_2v6     ;Set gr_2v6 flag
0062: C3                                clr c
0063: EE                                mov a,r6
0064: 9C                                subb a,r4
0065: 404C                              jc Small_2v6    ;<2.6V: clear flag, exhange registers
0067: 7004                              jnz Calculate
0069: EF                                mov a,r7
006A: 9D                                subb a,r5
006B: 4046                              jc Small_2v6
                                                        ;Calculate T2/T1
                                                        ;[R6.R7]=[R6.R7]/[R4.R5]
                                                        ; 1<Result<4.4
                                                        ;Before dividing: [R6.R7]=16*[R6.R7]
006D: 7806              Calculate:      mov r0,#06      ;r0 points to r6
006F: EF                                mov a,r7
0070: C4                                swap a
0071: 54F0                              anl a,#0f0h
0073: CF                                xch a,r7        ;r7 shifted 4 bits left
0074: C4                                swap a
0075: CE                                xch a,r6
0076: C4                                swap a          ;nibbles r6 swapped
0077: D6                                xchd a,@r0
0078: FE                                mov r6,a        ;r6 also shifted 4 bits to the left
```

```
TSW-ASM51 V3.0b Serial #00052252  Main program
PAGE   4

LOC   OBJ         LINE  SOURCE

0079: 120000   R  127                lcall __sdivi   ;[R6.R7]=16*T2/T1
                  128
007C: 300105   R  129                jnb Gr_2v6,Sec_tab

007F: 900000   R  130                mov dptr,#Seg_tab_1
0082: 8003        131                sjmp Calc_point
0084: 900000   R  132  Sec_tab:      mov dptr,#Seg_tab_2
0087: C3          133  Calc_point:   clr c
0088: EF          134                mov a,r7        ;Calculate address of segment info
0089: 33          135                rlc a           ;2*R7
008A: f582        136                mov dpl,a
008C: C3          137                clr c
008D: 9494        138                subb a,#148     ;If a>148 then V>5.0: Display H.I V
008F: 4003        139                jc Display
0091: 758294      140                mov dpl,#148    ;V>5.0V
                  141
0094:             142  Display:                      ;Display result.DPTR points to segment data
0094: C2AB        143                clr et1         ;Disable LCD switch interrupt
0096: E4          144                clr a
0097: 93          145                movc a,@a+dptr  ;Get segment_1 data (digit + LOBAT)
0098: 209509      146                jb LCD_COM,Com_1 ;If LCD_COM=1, then invert result
009B: F5B0        147                mov Digit_1,a
009D: 7401        148                mov a,#01
009F: 93          149                movc a,@a+dptr  ;Get segment_2 data (digit + point)
00A0: F5A0        150                mov Digit_2,a
00A2: 01AF     R  151                ajmp New_meas
00A4:             152  Com_1:
00A4: 64Ff        153                xrl a,#0ffh
00A6: F5B0        154                mov Digit_1,a
00A8: 7401        155                mov a,#01
00AA: 93          156                movc a,@a+dptr  ;Get segment_2 data (digit+point)
00AB: 64FF        157                xrl a,#0ffh
00AD: F5A0        158                mov Digit_2,a
00AF:             159  New_meas:
00AF: D2AB        160                setb et1        ;Enable LCD switch interrupt
00B1: 0130     R  161                ajmp Start_ADC
                  162
                  163
00B3:             164  Small_2v6:                    ;V<2.55V. Exchange registers
00B3: C201     R  165                clr Gr_2v6      ;Clear flag
00B5: CC          166                xch a,r4        ;Exchange [R4.R5] and [R6.R7]
00B6: CE          167                xch a,r6
00B7: FC          168                mov r4,a
00B8: CD          169                xch a,r5
00B9: CF          170                xch a,r7
00BA: FD          171                mov r5,a
00BB: 80B0        172                sjmp Calculate  ;Start calculation
```

```
TSW-ASM51 V3.0b Serial #00052252  Main program
PAGE   5

LOC   OBJ          LINE  SOURCE
                                          extrn code(__sdivi)
                         $EJECT
;***************************************************************************
                         ;A/D measurement interrupt routine  (INT2)
;***************************************************************************
----                                      CSEG AT 3Bh     ;INT2 vector (Voltage measurement)
003B: 020000   R                          ljmp ADC_Int
----                                      RSEG ADC
0000; C28C     R         ADC_Int:         clr tr0         ;Stop timer_0
0002: B200     R                          cpl ADC_Status
0004: 300009   R                          jnb ADC_Status,Stat_0
                         ;ADC_Status is '1'
0007: AC8C                                mov r4,th0       ;Get timer_0 values
0009: AD8A                                mov r5,tl0       ;T1
000B: 53E9FE                              anl ix1,#0FEh    ;INT2 on '0' input level
                                                           ;Check discharge Vd+1.3V
000E: 8003                                sjmp Leave_INT2
                         ;ADC_Status is '0'
0010: 43E901             Stat_0:          orl ix1,#01H     ;INT2 on '1' input level
0013: C2C0               Leave_INT2:      clr iq2          ;Clear interrupt flag
0015: 32                                  reti
                         $eject
;***************************************************************************
                         ;Timer_1 interrupt for inverting LCD signals
;***************************************************************************
----                                      CSEG AT 18h      ;Timer_1 interrupt (LCD switch)
001B: 020000   R                          ljmp LCD_int
----                                      RSEG LCD
0000: 63B0FF   R         LCD_int:         xrl Digit_1,#0ffh ;Invert digit_1
0003: B295                                cpl LCD_COM       ;Invert LCD_COM
0005: 63A0FF                              xrl Digit_2,#0ffh ;Invert digit_2
```

```
TSW–ASM51 V3.0b Serial #00052252  Main program
PAGE   6

LOC   OBJ         LINE  SOURCE
0008: 0A                                inc Dis_tim     ;Update discharge timer
0009: 32                                reti
                        $eject
;***************************************************************************
                        ;Power failure interrupt routine (INT4)
;***************************************************************************
––––                                    CSEG AT 04Bh    ;Power failure interrupt from PCF1252
004B: 020000   R                        ljmp P_Fail
––––                                    RSEG Power_F
0000: C291     R        P_Fail:         clr Anal_in     ;Ground analog input
0002: E4                                clr a
0003: F5B0                              mov Digit_1,a   ;Clear LCD
0005: C295                              clr LCD_COM
0007: F5A0                              mov Digit_2,a
0009: 80FE                              sjmp $          ;Exit only with RESET !!
                        $eject
```

```
TSW–ASM51 V3.0b Serial #00052252  Main program
PAGE   7

LOC   OBJ          LINE  SOURCE
                    241  ;Table with LCD segment data for V>=2.55V
                    242
                    243
––––                244                 rseg table_1
                    245
0000:           R   246  Seg_Tab_1:
0000:               247                 ds 32                             ;V<2.55V. (adr:0..31)
0020: DBFD          248                 db 0dbh,0fdh                      ;’2.6V LOBAT’
                                                                          ;(adr:32..33)
0022: DB87DB87      249                 db 0dbh,87h,0dbh,87h              ;’2.7V LOBAT’
                                                                          ;(Adr:34..37)
0026: DBFFDBFF      250                 db 0dbh,0ffh,0dbh,0ffh            ;’2.8V LOBAT’
                                                                          ;(adr:38..41)
002A: DBEFDBeF      251                 db 0dbh,0efh,0dbh,0efh            ;’2.9V LOBAT’
                                                                          ;(adr:42..45)
002E: 4FBF          252                 db 04fh,0bfh                      ;’3.0V’  (Adr:46..47)
0030: 4F864F86      253                 db 04fh,86h,04fh,86h,04fh,86h     ;’3.1V’  (Adr:48..53)
0034: 4F86
0036: 4FDB4FDB      254                 db 4fh,0dbh,4fh,0dbh              ;’3.2V’  (Adr:54..57)
003A: 4FCF4FCF      255                 db 4fh,0cfh,4fh,0cfh              ;’3.3V’  (Adr:58..61)
003E: 4FE64FE6      256                 db 4fh,0e6h,4fh,0e6h              ;’3.4V’  (Adr:62..65)
0042: 4FED4FEC      257                 db 4fh,0edh,4fh,0edh              ;’3.5V’  (Adr:66..69)
0046: 4FFD4FFD      258                 db 4fh,0fdh,4fh,0fdh,4fh,0fdh     ;’3.6V’  (Adr:70..75)
004A: 4FFD
004C: 4F874F87      259                 db 4fh,87h,4fh,87h                ;’3.7V’  (Adr:76..79)
0050: 4FFF4FFF      260                 db 4fh,0ffh,4fh,0ffh              ;’3.8V’  (Adr:80..83)
0054: 4FEF4FEF      261                 db 4fh,0efh,4fh,0efh,4fh,0efh     ;’3.9V’  (Adr:84..89)
0058: 4FEF
005A: 66BF66BF      262                 db 66h,0bfh,66h,0bfh              ;’4.0V’  (Adr:90..93)
005E: 66866686      263                 db 66h,86h,66h,86h                ;’4.1V’  (Adr:94..97)
0062: 66DB66DB      264                 db 66h,0dbh,66h,0dbh,66h,0dbh     ;’4.2V’  (Adr:98..103)
0066: 66DB
0068: 66CF66CF      265                 db 66h,0cfh,66h,0cfh              ;’4.3V’  (Adr:104..107)
006C: 66E666E6      266                 db 66h,0e6h,66h,0e6h,66h,0e6h     ;’4.4V’  (Adr:108..113)
0070: 66E6
0072: 66ED66ED      267                 db 66h,0edh,66h,0edh,66h,0edh     ;’4.5V’  (Adr:114..119)
0076: 66ED
0078: 66FD66FD      268                 db 66h,0fdh,66h,0fdh              ;’4.6V’  (Adr:120..123)
007C: 66876687      269                 db 66h,87h,66h,87h,66h,87h        ;’4.7V’  (Adr:124..129)
0080: 6687
0082: 66FF66FF      270                 db 66h,0ffh,66h,0ffh,66h,0ffh     ;’4.8V’  (Adr:130..135)
0086: 66FF
0088: 66EF66EF      271                 db 66h,0efh,66h,0efh              ;’4.9V’  (Adr:136..139)
008C: 6DBF6DBF      272                 db 05dh,0bfh,06dh,0fbh,06dh,0bfh;’5.0V’  (Adr:140..147)
0090: 6DBF
0092: 6DBF          273                 db 06dh,0bfh
0094: 6B0           274                 db 76h,0b0h                       ;’H.IV’  (Adr:148..149)
                    275
                    276  $EJECT
```

```
TSW-ASM51 V3.0b Serial #00052252  Main program
PAGE   8

LOC   OBJ          LINE  SOURCE

                    277
                    278
;****************************************************************************
                    279  ;Table with LCD segment data for V<=2.55V
                    280
;****************************************************************************
                    281
----                282                rset table_2
                    283
0000:            R  284  Seg_Tab_2:
0000:               285                ds 32                             ;V>2.55V. (Adr:0..31)
0020: DBEDDBED      286                db 0dbh,0edh,0dbh,0edh,0dbh,0edh;'2.5V LOBAT'
0024: DBED                                                               ;(Adr:32..37)
0026: DBE6DBE6      287                db 0dbh,0e6h,0dbh,0e6h,0dbh,0e6h;'2.4V LOBAT'
002A: DBE6                                                               ;(Adr:38..43)
002C: DBCFDBCF      288                db 0dbh,0cfh,0dbh,0cfh,0dbh,0cfh;'2.3V LOBAT'
0030: DBCF                                                               ;(Adr:44..49)
0032: DBDBDBDB      289                db 0dbh,0dbh,0dbh,0dbh,0dbh,0dbh;'2.2V LOBAT'
0036: DBDB                                                               ;(Adr:50..57)
0038: DBDB          290                db 0dbh,0dbh
003A: DB86DB86      291                db 0dbh,86h,0dbh,86h,0dbh,86h   ;'2.1V LOBAT'
003E: DB86                                                               ;(Adr:58..69)
0040: DB86DB86      292                db 0dbh,86h,0dbh,86h,0dbh,86h
0044: DB86
0046: DBBFDBBF      293                db 0dbh,0bfh,0dbh,0bfh,0dbh,0bfh;'2.0V LOBAT'
004A: DBBF                                                               ;(Adr:70..85)
004C: DBBFDBBF      294                db 0dbh,0bfh,0dbh,0bfh,0dbh,0bfh
0050: DBBF
0052: DBBFDBBF      295                db 0dbh,0bfh,0dbh,0bfh
0056: 86EF86EF      296                db 86h,0efh,86h,0efh,86h,0efh   ;'1.9V LOBAT'
005A: 86EF                                                               ;(Adr:86..109)
005C: 86EF86EF      297                db 86h,0efh,86h,0efh,86h,0efh
0060: 86EF
0062: 86EF86EF      298                db 86h,0efh,86h,0efh,86h,0efh
0066: 86EF
0068: 86EF86EF      299                db 86h,0efh,86h,0efh,86h,0efh
006C: 86EF
006E: 86FF86FF      300                db 86h,0ffh,86h,0ffh,86h,0ffh   ;'1.8V LOBAT'
0072: 86FF                                                               ;(Adr:110..121)
0074: 86FF86FF      301                db 86h,0ffh,86h,0ffh,86h,0ffh
0078: 86FF
007A: 38BF38BF      302                db 38h,0bfh,38h,0bfh,38h,0bfh   ;'L.OV LOBAT'
007E: 38BF                                                               ;(Adr:122..133)
0080: 38BF38BF      303                db 38h,0bfh,38h,0bfh,38h,0bfh
0084: 38BF
                    304
0086:               305                end
```

A/D conversion with P83CL410 PCF1252-x

EIE/AN91006

Philips Semiconductors and Philips Electronics North America Corporation reserve the right to make changes, without notice, in the products, including circuits, standard cells, and/or software, described or contained herein in order to improve design and/or performance. Philips Semiconductors assumes no responsibility or liability for the use of any of these products, conveys no license or title under any patent, copyright, or mask work right to these products, and makes no representations or warranties that these products are free from patent, copyright, or mask work right infringement, unless otherwise specified. Applications that are described herein for any of these products are for illustrative purposes only. Philips Semiconductors makes no representation or warranty that such applications will be suitable for the specified use without further testing or modification.

LIFE SUPPORT APPLICATIONS
Philips Semiconductors and Philips Electronics North America Corporation Products are not designed for use in life support appliances, devices, or systems where malfunction of a Philips Semiconductors and Philips Electronics North America Corporation Product can reasonably be expected to result in a personal injury. Philips Semiconductors and Philips Electronics North America Corporation customers using or selling Philips Semiconductors and Philips Electronics North America Corporation Products for use in such applications do so at their own risk and agree to fully indemnify Philips Semiconductors and Philips Electronics North America Corporation for any damages resulting from such improper use or sale.

Philips Semiconductors
811 East Arques Avenue
P.O. Box 3409
Sunnyvale, California 94088–3409
Telephone 800-234-7381

© Copyright Philips Electronics North America Corporation 1997
All rights reserved. Printed in U.S.A.